
1 Overview

This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2019-06_D20wk25 related to the title of this CR.

Contact:
Shaun Harry

shaun.harry@keysight.com
2 Corrections required
2.1 Change 1
	Function name
	fl_TC_10_1_6_1_Body

	Reason for change
	At Step 7, the PTI in PDU SESSION RELEASE COMMAND should match the  value recieved in PDU SESSION RELEASE REQUEST at Step 4.


	Summary of change
	Use PTI indicated in PDU SESSION RELEASE REQUEST message for step 7

	TTCN module
	GSM_TestCases_NR5GC.ttcn

	MCC160 Comment
	Accpeted


Before Change:

	  function fl_TC_10_1_6_1_Body() runs on NR5GC_PTC

  {

    var NR_CellId_Type v_NGC_NASCellA := f_NR5GC_MapNASCell(ngc_CellA, NG_AllCellsOnSamePLMN);

    var PDUSessionInfoList_Type v_PDUSessionInfo;

    var O1_Type v_PDUSessionIdToRelease;

    var GSM_MobilityInfo_Type v_GSM_MobilityInfo;

    var NR_RadioBearerId_Type v_SRB;

    var template (value) NG_NAS_DL_Pdu_Type v_MsgToSend;

    var NR_SRB_COMMON_IND v_ReceivedAsp;

    var charstring v_NSSAI;

    timer t_Wait := 3.0; // @sic R5-195981 sic@

    v_SRB := tsc_NR_RbId_SRB2;

    v_NSSAI := f_ConvertS_NSSAI_ForATCommand (cs_S_NSSAI_SST1eMBB);

    //@siclog "Steps 1" siclog@

    f_UT_RequestPDUSession(UT, 3, tsc_RandomAPN, -, -, v_NSSAI);

    //@siclog "Steps 2" siclog@

    f_NR5GC_PDUSessionEstablishment (v_NGC_NASCellA, 1);

    //@siclog "Steps 3" siclog@

    //Cause the UE to transmit a PDU SESSION RELEASE REQ

    f_UT_DeactivateBearer(UT, 3);

    v_PDUSessionInfo := f_NR5GC_MobileInfo_GetSessionInfoList();

    v_PDUSessionIdToRelease := v_PDUSessionInfo[f_NR5GC_MobileInfo_GetNoOfSessions()-1].SessionId; // @sic R5s190629 sic@

    //@siclog "Steps 4" siclog@

    // The UE transmits a PDU SESSION RELEASE REQ

    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered)))) -> value v_ReceivedAsp; // @sic R5s190629 sic@

    if (not f_Check_NG_PDUSessionReleaseReq (v_GSM_MobilityInfo, v_PDUSessionInfo,

                                             v_ReceivedAsp.Signalling.Nas[0].Pdu, v_PDUSessionIdToRelease)) {  // @sic R5s190629, R5s200564 sic@

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");

    } else {

        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    }

    //@siclog "Steps 5" siclog@

    // send PDU Session Modification command

    v_MsgToSend := f_Get_NG_PDUSessionModificationCommand(v_PDUSessionIdToRelease);

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA, - , -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));

    //@siclog "Steps 6" siclog@

    // Check: Does the UE transmit a PDU SESSION MODIFICATION COMPLETE or PDU SESSION MODIFICATION COMMAND REJECT  ?

    t_Wait.start;

    alt

    {

        []SRB.receive (car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp

        {

            t_Wait.stop;

        }

        []t_Wait.timeout

        {

            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");

        }

    }

    if (match (v_ReceivedAsp.Signalling.Nas[0].Pdu.PiggybackedPduList[0].Msg , cr_NG_PDU_SESSION_MODIFICATION_COMPLETE(?,?))) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");

    } else if (match (v_ReceivedAsp.Signalling.Nas[0].Pdu.PiggybackedPduList[0].Msg, cr_NG_PDU_SESSION_MODIFICATION_COMMAND_REJECT(?,?))) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");

    }

    //@siclog "Steps 7" siclog@

    // send PDU Session Release Command

    v_MsgToSend := f_Get_NG_PDUSessionReleaseCommand(v_PDUSessionIdToRelease, cs_GMM_GSM_Cause (omit, '00100100'B));
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA, - , -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));

    //@siclog "Steps 8" siclog@

    // The UE transmits a PDU SESSION RELEASE Complete

    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    if (not f_Check_NG_PDUSessionReleaseComplete (v_PDUSessionInfo, v_ReceivedAsp.Signalling.Nas[0].Pdu, v_PDUSessionIdToRelease)) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 8");

    }

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8");

    } //End of test body of TC_10_1_6_1



After Change:

	function fl_TC_10_1_6_1_Body() runs on NR5GC_PTC

  {

    var NR_CellId_Type v_NGC_NASCellA := f_NR5GC_MapNASCell(ngc_CellA, NG_AllCellsOnSamePLMN);

    var PDUSessionInfoList_Type v_PDUSessionInfo;

    var O1_Type v_PDUSessionIdToRelease;

    var GSM_MobilityInfo_Type v_GSM_MobilityInfo;

    var NR_RadioBearerId_Type v_SRB;

    var template (value) NG_NAS_DL_Pdu_Type v_MsgToSend;

    var NR_SRB_COMMON_IND v_ReceivedAsp;

    var charstring v_NSSAI;

    var ProcedureTransactionIdentifier v_PTI; // KS CR R5s200xxx
    timer t_Wait := 3.0; // @sic R5-195981 sic@

    v_SRB := tsc_NR_RbId_SRB2;

    v_NSSAI := f_ConvertS_NSSAI_ForATCommand (cs_S_NSSAI_SST1eMBB);

    //@siclog "Steps 1" siclog@

    f_UT_RequestPDUSession(UT, 3, tsc_RandomAPN, -, -, v_NSSAI);

    //@siclog "Steps 2" siclog@

    f_NR5GC_PDUSessionEstablishment (v_NGC_NASCellA, 1);

    //@siclog "Steps 3" siclog@

    //Cause the UE to transmit a PDU SESSION RELEASE REQ

    f_UT_DeactivateBearer(UT, 3);

    v_PDUSessionInfo := f_NR5GC_MobileInfo_GetSessionInfoList();

    v_PDUSessionIdToRelease := v_PDUSessionInfo[f_NR5GC_MobileInfo_GetNoOfSessions()-1].SessionId; // @sic R5s190629 sic@

    //@siclog "Steps 4" siclog@

    // The UE transmits a PDU SESSION RELEASE REQ

    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered)))) -> value v_ReceivedAsp; // @sic R5s190629 sic@

    if (not f_Check_NG_PDUSessionReleaseReq (v_GSM_MobilityInfo, v_PDUSessionInfo,

                                             v_ReceivedAsp.Signalling.Nas[0].Pdu, v_PDUSessionIdToRelease)) {  // @sic R5s190629, R5s200564 sic@

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");

    } else {

        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    }

       v_PTI := v_ReceivedAsp.Signalling.Nas[0].Pdu.PiggybackedPduList[0].Msg.pdu_Session_Release_Request.procedureTransactionIdentifier;
    //@siclog "Steps 5" siclog@

    // send PDU Session Modification command

    v_MsgToSend := f_Get_NG_PDUSessionModificationCommand(v_PDUSessionIdToRelease);

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA, - , -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));

    //@siclog "Steps 6" siclog@

    // Check: Does the UE transmit a PDU SESSION MODIFICATION COMPLETE or PDU SESSION MODIFICATION COMMAND REJECT  ?

    t_Wait.start;

    alt

    {

        []SRB.receive (car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp

        {

            t_Wait.stop;

        }

        []t_Wait.timeout

        {

            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");

        }

    }

    if (match (v_ReceivedAsp.Signalling.Nas[0].Pdu.PiggybackedPduList[0].Msg , cr_NG_PDU_SESSION_MODIFICATION_COMPLETE(?,?))) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");

    } else if (match (v_ReceivedAsp.Signalling.Nas[0].Pdu.PiggybackedPduList[0].Msg, cr_NG_PDU_SESSION_MODIFICATION_COMMAND_REJECT(?,?))) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");

    }

    //@siclog "Steps 7" siclog@

    // send PDU Session Release Command
    v_MsgToSend := f_Get_NG_PDUSessionReleaseCommand(v_PDUSessionIdToRelease, cs_GMM_GSM_Cause(omit, '00100100'B),-,-,-,-,v_PTI ); // 5GSM cause "Regular deactivation"
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA, - , -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));

    //@siclog "Steps 8" siclog@

    // The UE transmits a PDU SESSION RELEASE Complete

    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    if (not f_Check_NG_PDUSessionReleaseComplete (v_PDUSessionInfo, v_ReceivedAsp.Signalling.Nas[0].Pdu, v_PDUSessionIdToRelease)) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 8");

    }

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8");

    } //End of test body of TC_10_1_6_1  


2.2 Change 2
	Function name
	f_Get_NG_PDUSessionReleaseCommand

	Reason for change
	Current implementation has no option to specify ProcedureTransactionIdentifier.

	Summary of change
	Added new parameter for PTI

	TTCN module
	NG_NASTemplateFunctions.ttcn

	MCC160 Comment
	Accepted


Before Change:

	function f_Get_NG_PDUSessionReleaseCommand(template (value) O1_Type p_PDU_SessionId,

                                             template (value) GMM_GSM_Cause p_Cause := cs_GMM_GSM_Cause (omit, '00011010'B),

                                             template (omit) GPRS_Timer3 p_BackOff := omit,

                                             template (omit) NGSM_CongestionReattemptInd p_NGSM_CongestionReattemptInd := omit, // @sic R5s190543 sic@,

                                             template (omit) ExtdProtocolConfigOptions p_ExtdPCO := omit,

                                             template (omit) EAP_Message p_EAP := omit) return template (value) NG_NAS_DL_Pdu_Type

  {

    var template (value) NG_NAS_DL_Message_Type v_GSMMsg;

    var ProcedureTransactionIdentifier v_PTI := '00'O;
    v_GSMMsg := cs_NG_PDU_SESSION_RELEASE_COMMAND (cs_NG_PDU_SessionId (p_PDU_SessionId, omit), // @sic R5s200564 sic@

                                                   v_PTI,
                                                   p_Cause,

                                                   p_BackOff,

                                                   p_EAP,

                                                   p_NGSM_CongestionReattemptInd,

                                                   p_ExtdPCO);

    return f_Get_NG_DLNASTransport (cs_PayloadContainerType(tsc_PayloadContainerESMMsg), // @sic R5w190208 sic@

                                    v_GSMMsg,

                                    -,

                                    p_PDU_SessionId, // @sic R5s190626, R5s200564 sic@

                                    omit,

                                    omit,

                                    p_BackOff);

  }


After Change:

	 function f_Get_NG_PDUSessionReleaseCommand(template (value) O1_Type p_PDU_SessionId,

                                             template (value) GMM_GSM_Cause p_Cause := cs_GMM_GSM_Cause (omit, '00011010'B),

                                             template (omit) GPRS_Timer3 p_BackOff := omit,

                                             template (omit) NGSM_CongestionReattemptInd p_NGSM_CongestionReattemptInd := omit, // @sic R5s190543 sic@,

                                             template (omit) ExtdProtocolConfigOptions p_ExtdPCO := omit,

                                             template (omit) EAP_Message p_EAP := omit,

                                             template (value) ProcedureTransactionIdentifier p_PTI := '00'O) return template (value) NG_NAS_DL_Pdu_Type

  {

    var template (value) NG_NAS_DL_Message_Type v_GSMMsg;

    v_GSMMsg := cs_NG_PDU_SESSION_RELEASE_COMMAND (cs_NG_PDU_SessionId (p_PDU_SessionId, omit), // @sic R5s200564 sic@

                                                   p_PTI,

                                                   p_Cause,

                                                   p_BackOff,

                                                   p_EAP,

                                                   p_NGSM_CongestionReattemptInd,

                                                   p_ExtdPCO);

    return f_Get_NG_DLNASTransport (cs_PayloadContainerType(tsc_PayloadContainerESMMsg), // @sic R5w190208 sic@

                                    v_GSMMsg,

                                    -,

                                    p_PDU_SessionId, // @sic R5s190626, R5s200564 sic@

                                    omit,

                                    omit,

                                    p_BackOff);

  }
 


